
Subject: Re: pth order auto-regressive process with a specified mean and variance
Posted by [Yngvar Larsen](#) on Mon, 13 Feb 2012 15:04:05 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Feb 13, 9:19 am, Tom Van Niel <mookieth...@gmail.com> wrote:

> Hi Guys,
>
> Does anybody have IDL code that simulates a pth order auto-regressive
> process with a specified mean and variance? If so, please let me
> know.
>

Well. Assuming that the parameters of your AR(p) process is chosen such that the process is wide-sense stationary, this should be easy:

```
xmean = 0.9
xvariance = 1.2
;; This example vector of AR(4) coefficients
;; results in a WSS process.
;; Homework: make sure this is the case yourself.
;; (Hint: Roots of characteristic polynomial
;;       within unit circle)
;; My model:
;;  $X_t = \sum_{i=1}^p a_i X_{t-i} + n_t$ 
;; with  $n_t$  iid normal.
;; Array A below contains (after reverse)
;;  $A = [a_{p-1}, \dots, a_2, a_1]$ 
A = reverse([2.7607, -3.8106, 2.6535, -0.9238])
p = n_elements(A)
transient = 1000 ; Transient throwaway points
npoints = 10000 + transient

drive_proc = sqrt(xvariance)*randomn(seed, npoints)
ar_proc = fltarr(npoints)
ar_proc[0] = drive_proc[0]
for ii=1, p-1 do $
  ar_proc[ii] = drive_proc[ii] + $
  total(A[p-ii:]*ar_proc[0:ii-1])

for ii=p, npoints-1 do $
  ar_proc[ii] = drive_proc[ii] + $
  total(A*ar_proc[ii-p:ii-1])

;; Remove transient points, where
;; the process isn't WSS yet.
;; Add mean value.
ar_proc = ar_proc[transient:*] + xmean
```

--
Yngvar
